
Μια γραµµατική για τα Αγγλικά

Μια πρόταση αποτελείται 
από µια ουσιαστική φράση, 
ένα ρήµα, και µια ουσιαστική 
φράση

Μια ουσιαστική φράση 
αποτελείται από ένα άρθρο 

<S> ::= <NP> <V> <NP>

<NP> ::= <A> <N>

αποτελείται από ένα άρθρο 
και ένα ουσιαστικό

Ρήµατα είναι τα εξής…

Άρθρα είναι τα εξής…

Ουσιαστικά είναι τα εξής...

<NP> ::= <A> <N>

<V> ::= loves | hates | eats

<A> ::= a | the

<N> ::= dog | cat | rat

Πως δουλεύει µια γραµµατική

• Μια γραµµατική είναι ένα σύνολο κανόνων που ορίζουν 

το πως κατασκευάζεται ένα συντακτικό δένδροσυντακτικό δένδροσυντακτικό δένδροσυντακτικό δένδρο

• Ξεκινάµε βάζοντας το <S> στη ρίζα του δένδρου

• Οι κανόνες της γραµµατικής λένε πως µπορούµε να 

προσθέσουµε παιδιά σε κάθε σηµείο του δένδρουπροσθέσουµε παιδιά σε κάθε σηµείο του δένδρου

• Για παράδειγµα, ο κανόνας

<S> ::= <NP> <V> <NP>

λέει ότι µπορούµε να προσθέσουµε κόµβους <NP>, <V>, 

και <NP>, µε αυτή τη σειρά, ως παιδιά του κόµβου <S>

Γραµµατική για αριθµητικές εκφράσεις

<expr> ::= <expr> + <expr> 

| <expr> * <expr> 

| ( <expr> )

| a | b | c

• Μια αριθµητική έκφραση µπορεί να είναι

– το άθροισµα δύο άλλων εκφράσεων, ή

– το γινόµενο δύο εκφράσεων, ή

– µια έκφραση που περικλείεται από παρενθέσεις, ή

– κάποια από τις µεταβλητές a, b, ή c

| a | b | c



Συντακτικό δένδρο

<expr>

( <expr> )

<expr> * <expr>

<expr> + <expr>

<expr> * <expr>

( <expr> )

a b

c

Συστατικά µιας γραµµατικής

<S> ::= <NP> <V> <NP>

<NP> ::= <A> <N>

αρχικό σύµβολο

κανόνας
παραγωγής

<V> ::= loves | hates|eats

<A> ::= a | the

<N> ::= dog | cat | rat

µη τερµατικά 
σύµβολα

τερµατικά σύµβολα
(λεκτικές µονάδες)

Ορισµός γραµµατικών σε µορφή Backus-Naur

• Μια γραµµατική σε µορφή Backus-Naur αποτελείται από

– Ένα σύνολο από λεκτικές µονάδες λεκτικές µονάδες λεκτικές µονάδες λεκτικές µονάδες (tokenstokenstokenstokens)

• Συµβολοσειρές που αποτελούν τα µικρότερα αδιαίρετα κοµµάτια 
της σύνταξης του προγράµµατος

– Ένα σύνολο από µη τερµατικά σύµβολατερµατικά σύµβολατερµατικά σύµβολατερµατικά σύµβολα (nonnonnonnon----terminalsterminalsterminalsterminals)

• Συµβολοσειρές που εγκλείονται σε αγκύλες, π.χ. <NP>, και 
αντιπροσωπεύουν κοµµάτια του συντακτικού της γλώσσας αντιπροσωπεύουν κοµµάτια του συντακτικού της γλώσσας 

• ∆ε συναντιούνται στο πρόγραµµα, αλλά είναι σύµβολα που 
βρίσκονται στο αριστερό µέρος κάποιων κανόνων της γραµµατικής

– Το αρχικό σύµβολο αρχικό σύµβολο αρχικό σύµβολο αρχικό σύµβολο (start symbolstart symbolstart symbolstart symbol) της γραµµατικής

• Ένα συγκεκριµένο µη τερµατικό σύµβολο που αποτελεί τη ρίζα του 
συντακτικού δένδρου για κάθε αποδεκτό από τη γλώσσα πρόγραµµα

– Ένα σύνολο από κανόνες παραγωγήςκανόνες παραγωγήςκανόνες παραγωγήςκανόνες παραγωγής (production rulesproduction rulesproduction rulesproduction rules)

Κανόνες παραγωγής

• Οι κανόνες παραγωγής χρησιµοποιούνται για την 

κατασκευή του συντακτικού δένδρου

• Κάθε κανόνας έχει τη µορφή Α ::= ∆

– Το αριστερό µέρος Α αποτελείται από ένα µη τερµατικό σύµβολο 

– Το δεξί µέρος ∆ είναι µια ακολουθία από τερµατικά (λεκτικές – Το δεξί µέρος ∆ είναι µια ακολουθία από τερµατικά (λεκτικές 
µονάδες) και µη τερµατικά σύµβολα

• Κάθε κανόνας προσδιορίζει έναν πιθανό τρόπο 

κατασκευής του συντακτικού υποδένδρου που

– έχει ως ρίζα του το µη τερµατικό σύµβολο στο αριστερό µέρος Α 

του κανόνα  και

– ως παιδιά αυτής της ρίζας (µε την ίδια σειρά εµφάνισης) έχει τα 
σύµβολα στο δεξί µέρος ∆ του κανόνα

Επιλογές στη γραφή των κανόνων παραγωγής

• Όταν υπάρχουν περισσότεροι από ένας κανόνες 

παραγωγής µε το ίδιο αριστερό µέρος, µπορούµε να 

κάνουµε χρήση της παρακάτω συντοµογραφίας

• Στη BNF γραµµατική µπορούµε να δώσουµε το αριστερό 

µέρος, το διαχωριστή ::=, και µετά µια ακολουθία από µέρος, το διαχωριστή ::=, και µετά µια ακολουθία από 

δεξιά µέρη που διαχωρίζονται από το ειδικό σύµβολο |

• Οι δύο γραµµατικές είναι ίδιες

<expr> ::= <expr> + <expr> 

| <expr> * <expr> 

| ( <expr> )

| a | b | c

<expr> ::= <expr> + <expr> 

<expr> ::= <expr> * <expr> 

<expr> ::= ( <expr> )

<expr> ::= a

<expr> ::=  b

<expr> ::= c

Κανόνες παραγωγής του κενού

• Το ειδικό µη τερµατικό <empty> χρησιµοποιείται σε 

περιπτώσεις που θέλουµε κάποιος κανόνας να µην 

παράγει τίποτα

• Για παράδειγµα, οι παρακάτω κανόνες ορίζουν τη δοµή 

if-then των περισσοτέρων γλωσσών, η οποία επιτρέπει if-then των περισσοτέρων γλωσσών, η οποία επιτρέπει 

την ύπαρξη ενός προαιρετικού else

<if-stmt> ::= if <expr> then <stmt> <else-part>
<else-part> ::= else <stmt> | <empty>

Κατασκευή συντακτικών δένδρων

• Αρχίζουµε την κατασκευή βάζοντας το αρχικό σύµβολο 

της γραµµατικής στη ρίζα του δένδρου

• Προσθέτουµε παιδιά σε κάθε µη τερµατικό σύµβολο,

χρησιµοποιώντας κάποιον από τους κανόνες παραγωγής

της γλώσσας για το συγκεκριµένο µη τερµατικότης γλώσσας για το συγκεκριµένο µη τερµατικό

• Η διαδικασία τερµατίζει όταν όλα τα φύλλα του δένδρου 

αποτελούνται από λεκτικές µονάδες

• Η συµβολοσειρά που αντιστοιχεί στο δένδρο βρίσκεται 

διαβάζοντας τα φύλλα του δένδρου από αριστερά προς 

τα δεξιά

Παραδείγµατα

• Τα συντακτικά δένδρα για τις παρακάτω εκφράσεις

a+b (a+b) (a+(b)) a*b+c

• Η κατασκευή των συντακτικών δένδρων είναι η δουλειά 

του συντακτικού αναλυτή ενός µεταγλωττιστή

• Υπάρχουν διάφοροι αποδοτικοί αλγόριθµοι και εργαλεία 

για ηµιαυτόµατη κατασκευή του συντακτικού αναλυτή



Τυπικός ορισµός σύνταξης γλωσσών

• Για να ορίσουµε τη σύνταξη των γλωσσών 

προγραµµατισµού χρησιµοποιούµε γραµµατικές

• Η γλώσσα που ορίζεται από µια γραµµατική είναι το 

σύνολο των συµβολοσειρών για τα οποία η γραµµατική 

µπορεί να παράξει συντακτικά δένδραµπορεί να παράξει συντακτικά δένδρα

• Τις περισσότερες φορές το σύνολο αυτό είναι άπειρο 

(παρόλο που η γραµµατική είναι πεπερασµένη)

• Η κατασκευή µιας γραµµατικής για µια γλώσσα µοιάζει 

λίγο µε προγραµµατισµό...

Παράδειγµα κατασκευής γραµµατικής (1)

Συνήθως γίνεται µε χρήση της τεχνικής “διαίρει και 

βασίλευε” (divide and conquer)

• ΠαράδειγµαΠαράδειγµαΠαράδειγµαΠαράδειγµα: κατασκευή της γλώσσας των δηλώσεων της 

Java:Java:

– αρχικά, η δήλωση έχει ένα όνοµα τύπου

– στη συνέχεια µια ακολουθία από µεταβλητές που διαχωρίζονται µε 
κόµµατα (όπου κάθε µεταβλητή µπορεί να πάρει µια αρχική τιµή)

– και στο τέλος ένα ερωτηµατικό (semicolon)

float a;
boolean a, b, c;
int a = 1, b, c = 1 + 2;

Παράδειγµα κατασκευής γραµµατικής (2)

• Αρχικά ας αγνοήσουµε την πιθανή ύπαρξη αρχικοποιητών:

• Ο κανόνας για τα ονόµατα των πρωτόγονων τύπων 

(primitive types) της Java είναι απλούστατος:

<type-name> ::= boolean | byte | short | int

<var-decl> ::= <type-name> <declarator-list> ;

ΣηµείωσηΣηµείωσηΣηµείωσηΣηµείωση:::: δεν παίρνουµε υπόψη κατασκευασµένους τύπους (constructed 

types): ονόµατα κλάσεων, ονόµατα διεπαφών (interfaces), και τύπους 

πινάκων

<type-name> ::= boolean | byte | short | int
| long | char | float | double

Παράδειγµα κατασκευής γραµµατικής (3)

• Η ακολουθία των µεταβλητών που διαχωρίζονται µε 

κόµµατα έχει ως εξής:

• Όπου ξανά, έχουµε προς το παρόν αγνοήσει τους 

<declarator-list> ::= <declarator> 
| <declarator> , <declarator-list>

• Όπου ξανά, έχουµε προς το παρόν αγνοήσει τους 

πιθανούς αρχικοποιητές των µεταβλητών

Παράδειγµα κατασκευής γραµµατικής (4)

• Οι δηλωτές µεταβλητών, µε ή χωρίς αρχικοποιήσεις, 

ορίζονται ως:

<declarator> ::= <variable-name> 
| <variable-name> = <expr>

• Για ολόκληρη τη Java:

– Πρέπει να επιτρέψουµε και ζεύγη από αγκύλες µετά το όνοµα 
των µεταβλητών για τη δήλωση των πινάκων

– Πρέπει επίσης να ορίσουµε και τη σύνταξη των αρχικοποιητών 
πινάκων

– (Φυσικά θέλουµε και ορισµούς για τα µη τερµατικά σύµβολα 
<variable-name> και <expr>)

Τί αποτελεί λεκτική µονάδα;

• Όποια κοµµάτια της γλώσσας επιλέξουµε να θεωρήσουµε 

ως µη κατασκευαζόµενα από µικρότερα κοµµάτια

• Μεταβλητές (i), λέξεις κλειδιά (if), τελεστές (==,++), 

σταθερές (123.4), κ.λπ.

• Οι γραµµατικές που έχουµε ορίσει δίνουν τη δοµή των • Οι γραµµατικές που έχουµε ορίσει δίνουν τη δοµή των 

φράσεων (phrase structure): πως το πρόγραµµα 

κατασκευάζεται από µια σειρά λεκτικών µονάδων

• Πρέπει επιπλέον να ορίσουµε και τη λεκτική δοµή (lexical 

structure): πως ένα αρχείο χωρίζεται σε λεκτικές µονάδες

∆ύο επιλογές ορισµού της λεκτικής δοµής

• Με µια κοινή γραµµατική

– Οι χαρακτήρες είναι οι µοναδικές λεκτικές µονάδες

– Συνήθως δεν ακολουθείται αυτή η επιλογή: κενά και σχόλια 
περιπλέκουν αρκετά τη γραµµατική και την καθιστούν µη 
αναγνώσιµη

• Με ξεχωριστές γραµµατικές• Με ξεχωριστές γραµµατικές

1. Μία που ορίζει πώς προκύπτουν οι λεκτικές µονάδες από ένα 
αρχείο µε χαρακτήρες

• Η γραµµατική αυτή συνήθως είναι µια κανονική γραµµατική (regular 
grammar) και χρησιµοποιείται από το λεκτικό αναλυτήλεκτικό αναλυτήλεκτικό αναλυτήλεκτικό αναλυτή ((((scannerscannerscannerscanner) ) ) ) 

2. Μία που ορίζει πως προκύπτουν τα συντακτικά δένδρα από µία 
ακολουθία λεκτικών µονάδων

• Η γραµµατική αυτή συνήθως είναι µια γραµµατική ελεύθερη 
συµφραζοµένων (context-free grammar) και χρησιµοποιείται από το 
συντακτικό αναλυτή (συντακτικό αναλυτή (συντακτικό αναλυτή (συντακτικό αναλυτή (parserparserparserparser))))

Ιστορικές σηµειώσεις (1)

• Παλιά, κάποιες γλώσσες προγραµµατισµού δε 

διαχώριζαν τη λεκτική από την φραστική δοµή

– Παλιές εκδόσεις της Fortran και της Algol επέτρεπαν κενά σε 
οποιοδήποτε σηµείο, ακόµα και στο µέσο µιας λέξης κλειδί!

– Άλλες γλώσσες, π.χ. η PL/I, επιτρέπουν τη χρήση λέξεων 
κλειδιών ως µεταβλητέςκλειδιών ως µεταβλητές

• (Το ίδιο συµβαίνει και στην ML, αλλά εκεί δεν αποτελεί πρόβληµα.)

• Τα παραπάνω

– προσθέτουν δυσκολία στην λεκτική και συντακτική ανάλυση και

– µειώνουν την αναγνωσιµότητα των προγραµµάτων

Ιστορικές σηµειώσεις (2)

• Κάποιες γλώσσες έχουν λεκτική δοµή σταθερής µορφής λεκτική δοµή σταθερής µορφής λεκτική δοµή σταθερής µορφής λεκτική δοµή σταθερής µορφής 

(fixed-format )—τα κενά είναι σηµαντικά

– Μία εντολή ανά γραµµή (π.χ. της διάτρητης κάρτας)

– Οι πρώτες 7 θέσεις κάθε γραµµής για την ταµπέλα (label)

• Οι πρώτες διάλεκτοι της Fortran, Cobol, και της Basic

• Σχεδόν οι περισσότερες µοντέρνες γλώσσες είναι 

ελεύθερης µορφήςελεύθερης µορφήςελεύθερης µορφήςελεύθερης µορφής (free-format ): τα κενά αγνοούνται

– Π.χ. Algol, Pascal, Java

– Μερικές άλλες (C, C++) διαφέρουν

λίγο λόγω του προεπεξεργαστή

Άλλες µορφές γραµµατικών

• Μικρές διαφοροποιήσεις της µορφής Backus-Naur (BNF)

– Χρήση  → ή = αντί για ::=

– Όχι <> αλλά κάποιο ειδικό font ή χρήση αποστρόφων για τις 
λεκτικές µονάδες ώστε να ξεχωρίζονται εύκολα από τα µη 
τερµατικά σύµβολα

• Επεκτάσεις της µορφής Backus-Naur (EBNF)• Επεκτάσεις της µορφής Backus-Naur (EBNF)

– Πρόσθετος συµβολισµός για την απλοποίηση κάποιων κανόνων:

• {x} υποδηλώνει µηδέν ή περισσότερες επαναλήψεις του x

• [x] υποδηλώνει ότι το x είναι προαιρετικό (δηλαδή x | <empty>)

• () για οµαδοποίηση

• | οπουδήποτε για να υποδηλώσει επιλογή

• Αποστρόφους γύρω από τις λεκτικές µονάδες ούτως ώστε να 
ξεχωρίζονται από τα παραπάνω µετασύµβολα

• Συντακτικά διαγράµµατα

Παραδείγµατα EBNF

<stmt-list> ::= {<stmt> ;}

<if-stmt> ::= if <expr> then <stmt> [else <stmt>]

<thing-list> ::= { (<stmt> | <declaration>) ;}

Συντακτικά διαγράµµατα (1)

• Έστω ότι έχουµε µια γραµµατική σε EBNF

• Ο κάθε κανόνας παραγωγής µετατρέπεται σε µια σειρά 

από κουτιά

– Ορθογώνια για τα µη τερµατικά σύµβολα

– Οβάλ για τα τερµατικά σύµβολα– Οβάλ για τα τερµατικά σύµβολα

– Τα παραπάνω ενώνονται µε βέλη

– (Πιθανώς κάποια βέλη να παρακάµπτουν κάποια από τα κουτιά.)

<if-stmt> ::= if <expr> then <stmt> [ else <stmt> ]

if then elseexpr stmt stmt
if-stmt



Συντακτικά διαγράµµατα (2)

• Πολλαπλοί κανόνες 

παραγωγής χρησιµοποιούν 

διακλαδώσεις (branching)

exp 

exp + exp 

exp * exp 

( exp ) 

a 

• Η επανάληψη υποδηλώνεται 

µε χρήση βρόχων

b 

c 

<exp> ::= <addend> {+ <addend>}

exp 
addend 

+ 

Παράδειγµα διαφορετικού συµβολισµού EBNF

WhileStatement:

while ( Expression ) Statement

DoStatement:

do Statement while ( Expression );

ForStatement:

for ( ForInitopt ; Expressionopt ; ForUpdateopt)

Statement

από το βιβλίο The Java™ Language Specification, James Gosling et al.

Από τη Σύνταξη προς τη Σηµασιολογία

Τρεις “ ισοδύναµες” γραµµατικές

<subexp> ::= a | b | c | <subexp> - <subexp>G1:

G3:

G2: <subexp> ::= <var> - <subexp> | <var>
<var> ::= a | b | c

<subexp> ::= <subexp> - <var> | <var>
<var> ::= a | b | c

• Και οι τρεις γραµµατικές ορίζουν την ίδια γλώσσα: τη 

γλώσσα όλων των συµβολοσειρών που περιλαµβάνουν 

ένα ή περισσότερα a, b, ή c τα οποία διαχωρίζονται από 

ένα µείον. Αλλά...

<var> ::= a | b | c



subexp  

<subexp> 

<var>   -  <subexp> 

c  

<var> 

<var>   -  <subexp> 

b  

a  G2 parse tree: 

<subexp> 

<subexp>  -   <var> 

a  

<var> 

<subexp>  -   <var> 

b  

c  G3 parse tree: 

Γιατί είναι σηµαντικά τα συντακτικά δένδρα;

• Θέλουµε η δοµή του συντακτικού δένδρου να 

αντικατοπτρίζει τη σηµασιολογία της συµβολοσειράς

που αντιπροσωπεύει

• Αυτό κάνει το σχεδιασµό της γλώσσας πιο δύσκολο:

– ενδιαφερόµαστε για τη δοµή του κάθε συντακτικού δένδρου– ενδιαφερόµαστε για τη δοµή του κάθε συντακτικού δένδρου

– όχι µόνο για την συµβολοσειρά των φύλλων του

Τα συντακτικά δένδρα είναι το µέρος που η σύνταξη 

αρχίζει να συναντά τη σηµασιολογία των γλωσσών

Τελεστές (operators)

• Τελεστές χρησιµοποιούνται για λειτουργίες που γίνονται 

συχνά, π.χ. πρόσθεση, αφαίρεση, πολλαπλασιασµό, …

• Ο όρος τελεστής αναφέρεται τόσο στη λεκτική µονάδα 

(π.χ. +, *) όσο και στη λειτουργία αυτή καθ’ αυτή

– ΜοναδιαίοιΜοναδιαίοιΜοναδιαίοιΜοναδιαίοι τελεστές δέχονται ένα όρισµα: -1– ΜοναδιαίοιΜοναδιαίοιΜοναδιαίοιΜοναδιαίοι τελεστές δέχονται ένα όρισµα: -1

– ∆υαδικοί∆υαδικοί∆υαδικοί∆υαδικοί τελεστές δέχονται δύο ορίσµατα: 1+2

– ΤριαδικοίΤριαδικοίΤριαδικοίΤριαδικοί τελεστές δέχονται τρία ορίσµατα: a?b:c

• Στις περισσότερες γλώσσες

– Οι δυαδικοί τελεστές γράφονται σε infix µορφή: π.χ. 1+2

– Οι µοναδιαίοι τελεστές γράφονται σε prefix (-2) ή σε postfix 
µορφή (i++)

Προτεραιότητα (precedence) τελεστών

• Έστω η γραµµατική G4: • Ένα συντακτικό δένδρο 

για τη συµβολοσειρά

a+b*c είναι το

<exp> ::= <exp> + <exp> 

| <exp> * <exp> 

| ( <exp> )

| a | b | c <exp> 

exp * exp  

<exp>   *   <exp> 

a  

<exp>   +   <exp> 

b  

c  

• Στο δένδρο αυτό η πρόσθεση γίνεται πριν από τον 

πολλαπλασιασµό, κάτι που δεν είναι σε αρµονία µε τις 

συνήθεις προτεραιότητες των τελεστών + και *



Προτεραιότητα τελεστών στη γραµµατική

• Για να έχουµε τη σωστή προτεραιότητα τελεστών, 

αλλάζουµε τη γραµµατική µε τέτοιο τρόπο ώστε ο 

τελεστής µε την µεγαλύτερη προτεραιότητα να 

καταλήγει “πιο κάτω” στο συντακτικό δένδρο

<exp> ::= <exp> + <exp> | <mulexp>
<mulexp> ::= <mulexp> * <mulexp>

| ( <exp> )
| a | b | c

G5:

Παραδείγµατα προτεραιότητας τελεστών

• C (15 επίπεδα προτεραιότητας—πάρα πολλά;)

• Pascal (5 επίπεδα—όχι αρκετά;)

a = b < c ? * p + b * c : 1 << d ()

a <= 0 or 100 <= a Συντακτικό λάθος!

• Smalltalk (1 επίπεδο για όλους τους δυαδικούς τελεστές)

a <= 0 or 100 <= a Συντακτικό λάθος!

a + b * c

Συντακτικό δένδρο µε σωστή προτεραιότητα

<exp> 

<exp>   +   <exp> 

 

G5 parse tree: <mulexp> <mulexp> 

Η γραµµατική G5 παράγει µόνο αυτό το δένδρο για a+b*c.

Αναγνωρίζει την ίδια γλώσσα µε τη G4, αλλά δεν παράγει 

πλέον δένδρα µε λάθος προτεραιότητα τελεστών.

c  

<mulexp> * <mulexp> 

b  

a  

Προσεταιριστικότητα (associativity) τελεστών

<exp> 

<exp>       +   <exp> 

<mulexp> 

<exp>   +   <exp> 

<mulexp> a  

<mulexp> 

<exp> 

<exp>    +      <exp> 

<mulexp> 

<exp>   +   <exp> 

<mulexp> c  

<mulexp> 

• Η γραµµατική G5 παράγει τα παραπάνω δένδρα για 

a+b+c. Το πρώτο από αυτά δεν αντικατοπτρίζει τη 

συνήθη προσεταιριστικότητα του τελεστή +.

c  b  b  a  



Προσεταιριστικότητα τελεστών

• Έχουν χρησιµότητα όταν η σειρά της αποτίµησης δεν 

καθορίζεται από παρενθέσεις ή προτεραιότητα τελεστών

• Οι αριστερά προσεταιριστικοί αριστερά προσεταιριστικοί αριστερά προσεταιριστικοί αριστερά προσεταιριστικοί τελεστές οµαδοποιούν από 

αριστερά προς τα δεξιά: a+b+c+d = ((a+b)+c)+d

• Οι δεξιά προσεταιριστικοί δεξιά προσεταιριστικοί δεξιά προσεταιριστικοί δεξιά προσεταιριστικοί τελεστές οµαδοποιούν από • Οι δεξιά προσεταιριστικοί δεξιά προσεταιριστικοί δεξιά προσεταιριστικοί δεξιά προσεταιριστικοί τελεστές οµαδοποιούν από 

δεξιά προς τα αριστερά:  a+b+c+d = a+(b+(c+d))

• Στις περισσότερες γλώσσες, οι περισσότεροι τελεστές

είναι αριστερά προσεταιριστικοί, αλλά υπάρχουν και 

εξαιρέσεις

Παραδείγµατα προσεταιριστικότητας

• C

• ML

a<<b<<c — τελεστές είναι αριστερά προσεταιριστικοί 
a=b=0 — δεξιά προσεταιριστικός (ανάθεση)

• Fortran

3-2-1 — τελεστές είναι αριστερά προσεταιριστικοί
1::2::nil — δεξιά προσεταιριστικός (κατασκευή λίστας)

a/b*c — τελεστές είναι αριστερά προσεταιριστικοί
a**b**c — δεξιά προσεταιριστικός (ύψωση σε δύναµη)

Προσεταιριστικότητα στη γραµµατική

• Για να διορθώσουµε το πρόβληµα, τροποποιούµε τη 

γραµµατική ώστε να κάνουµε δένδρα µε τελεστές να 

<exp> ::= <exp> + <exp> | <mulexp>
<mulexp> ::= <mulexp> * <mulexp>

| ( <exp> )
| a | b | c

G5:

γραµµατική ώστε να κάνουµε δένδρα µε τελεστές + να 

µεγαλώνουν προς τα αριστερά (παρόµοια για τον *)

G6: <exp> ::= <exp> + <mulexp> | <mulexp>
<mulexp> ::= <mulexp> * <rootexp> | <rootexp>
<rootexp> ::= ( <exp> )

| a | b | c

∆ένδρο µε σωστή προσεταιριστικότητα

<exp> 

<exp>    +     <mulexp> 

<rootexp> 

<exp>   +  <mulexp> 

<mulexp> c  

<rootexp> 

Η γραµµατική G6 παράγει µόνο αυτό το δένδρο για a+b+c.

Παράγει την ίδια γλώσσα µε τη G5, αλλά δεν παράγει 

πλέον δένδρα µε λάθος προσεταιριστικότητα τελεστών.

b  

<rootexp> <mulexp> c  

a  

<rootexp> 



Πρακτική άσκηση

• Ξεκινώντας από τη γραµµατική

G6: <exp> ::= <exp> + <mulexp> | <mulexp>
<mulexp> ::= <mulexp> * <rootexp> | <rootexp>
<rootexp> ::= ( <exp> )

| a | b | c

1. Προσθέστε έναν αριστερά προσεταιριστικό τελεστή &, 

µε προτεραιότητα µικρότερη από όλους τους άλλους

2. Στη συνέχεια προσθέστε ένα δεξιά προσεταιριστικό 

τελεστή **, µε προτεραιότητα µεγαλύτερη από όλους 

τους άλλους

Πολλαπλή ερµηνεία (ambiguity)

• Η γραµµατική G4 είναι διφορούµενη διφορούµενη διφορούµενη διφορούµενη (ambiguous)(ambiguous)(ambiguous)(ambiguous): 

παράγει περισσότερα από ένα συντακτικά δένδρα για 

την ίδια συµβολοσειρά

• Όµως η επιδιόρθωση των προβληµάτων προτεραιότητας 

και προσεταιριστικότητας των τελεστών εξαφάνισε όλη 

την ασάφεια στη συγκεκριµένη γραµµατικήτην ασάφεια στη συγκεκριµένη γραµµατική

• Αυτό είναι επιθυµητό: το συντακτικό δένδρο υποδηλώνει 

τη σηµασιολογία του προγράµµατος και δε θέλουµε 

αυτή να είναι διφορούµενη

• Όµως υπάρχουν και άλλοι λόγοι που µια γραµµατική είναι 

διφορούµενη, όχι µόνο λόγοι σχετικοί µε τους τελεστές

Το πρόβληµα του “ ξεκρέµαστου else”

<stmt> ::= <if-stmt> | s1 | s2
<if-stmt> ::= if <expr> then <stmt> else <stmt>

| if <expr> then <stmt>
<expr> ::= e1 | e2

• Η γραµµατική αυτή είναι διφορούµενη ως προς τo

“ξεκρέµαστο else” (“dangling-else ambiguity”).

• Η παρακάτω εντολή

if e1 then if e2 then s1 else s2

έχει δύο συντακτικά δένδρα

< if-stm t> 

if  <exp>  then  <stm t>  else  <stmt> 

e1  s2  

if   <exp>   then   <stmt> 

< if-stm t> 

e2  s1  

Οι περισσότερες γλώσσες

< if-stm t> 

if   <exp>   then   <stmt> 

e1  

s2  

if  <exp>  then  <stm t>  else  <stmt> 

< if-stm t> 

e2  s1  

Οι περισσότερες γλώσσες
που έχουν αυτό το πρόβληµα
επιλέγουν το κάτω συντακτικό

δένδρο: το else πηγαίνει µε το

κοντινότερο αταίριαστο then



∆ιόρθωση του προβλήµατος (1)

• Θέλουµε να επιβάλλουµε ότι εάν αυτό επεκταθεί σε ένα 

<stmt> ::= <if-stmt> | s1 | s2
<if-stmt> ::= if <expr> then <stmt> else <stmt>

| if <expr> then <stmt>
<expr> ::= e1 | e2

if, τότε το if πρέπει ήδη να έχει το δικό του else.  

Πρώτα, δηµιουργούµε ένα νέο µη τερµατικό <full-stmt> 

που παράγει όλες τις εντολές που παράγονται από το 

<stmt>, αλλά απαγορεύει τις if εντολές χωρίς else

<full-stmt> ::= <full-if> | s1 | s2
<full-if> ::= if <expr> then <full-stmt> else <full-stmt>

∆ιόρθωση του προβλήµατος (2)

• Μετά χρησιµοποιούµε το νέο µη τερµατικό εδώ

<stmt> ::= <if-stmt> | s1 | s2
<if-stmt> ::= if <expr> then <full-stmt> else <stmt>

| if <expr> then <stmt>
<expr> ::= e1 | e2

• Το αποτέλεσµα είναι ότι η παραπάνω γραµµατική µπορεί 

να ταιριάξει ένα else µε ένα if µόνο όταν όλα τα 

κοντινά if έχουν ήδη κάποιο else ως ταίρι τους.

Τώρα παράγουµε µόνο το συντακτικό δένδρο

<if-stmt> 

if   <exp>   then   <stmt> 

e1  <if-stmt> 

s2  

if  <exp>  then  <full-stmt>  else  <stmt> 

< >

e2  s1  

Ξεκρέµαστα else και αναγνωσιµότητα 

• ∆ιορθώσαµε τη γραµµατική, αλλά…

• Το πρόβληµα στη γραµµατική αντικατοπτρίζει κάποιο 

πρόβληµα στη γλώσσα, την οποία δεν αλλάξαµε

• Μια ακολουθία από if-then-else δεν είναι εύκολα 

αναγνώσιµη, ειδικά εάν κάποια από τα else λείπουναναγνώσιµη, ειδικά εάν κάποια από τα else λείπουν

int a=0;
if (0==0)
if (0==1) a=1;

else a=2;

Ποια είναι η τιµή του a
µετά την εκτέλεση του
προγράµµατος;



Καλύτερα στυλ προγραµµατισµού

int a=0;
if (0==0)
if (0==1) a=1;
else a=2;

Καλύτερο: σωστή στοίχιση

int a=0;
if (0==0) {
if (0==1) a=1;
else a=2;

}

Ακόµα καλύτερο: η χρήση
µπλοκ δείχνει καθαρά τη
δοµή του κώδικα

Γλώσσες χωρίς ξεκρέµαστα else

• Μερικές γλώσσες ορίζουν τα if-then-else µε τρόπο 

που επιβάλλει τη σαφήνεια στη χρήση τους

• Η Algol δεν επιτρέπει µέσα στο then να αρχίζει άµεσα 

κάποιο άλλο if

– (αλλά µπορεί να αρχίζει ένα µπλοκ µε ένα άλλο if)– (αλλά µπορεί να αρχίζει ένα µπλοκ µε ένα άλλο if)

• Η Algol 68 επιβάλλει σε κάθε if να τερµατίζεται µε ένα

fi (υπάρχουν επίσης do-od και case-esac)

• Η Ada επιβάλλει σε κάθε if να τερµατίζεται µε ένα

end if

Γραµµατική για ολόκληρη τη γλώσσα

• Κάθε ρεαλιστική γλώσσα περιέχει πολλά µη τερµατικά 

σύµβολα

• Ειδικά εάν από τη γραµµατική έχει εξαλειφθεί η ασάφεια

• Τα επιπλέον µη τερµατικά καθορίζουν τον τρόπο µε τον 

οποίο προκύπτει ένα µοναδικό συντακτικό δένδροοποίο προκύπτει ένα µοναδικό συντακτικό δένδρο

• Όταν κατασκευαστεί το συντακτικό δένδρο, τα επιπλέον 

µη τερµατικά δεν έχουν κάποια χρησιµότητα πλέον

• Οι υλοποιήσεις των γλωσσών συνήθως αποθηκεύουν µια 

συνεπτυγµένη µορφή του συντακτικού δένδρου, η οποία 

ονοµάζεται αφηρηµένο συντακτικό δένδροαφηρηµένο συντακτικό δένδροαφηρηµένο συντακτικό δένδροαφηρηµένο συντακτικό δένδρο

<exp> 

<exp>    +     <mulexp> 

<rootexp> 

<exp>   +  <mulexp> 

<mulexp> c  

<rootexp> 

Συγκεκριµένο συντακτικό δένδρο

+  

c  

a  

+  

b  

Αφηρηµένο συντακτικό δένδρο

b  

a  

<rootexp> 



Συµπερασµατικά

• Για τον ορισµό του συντακτικού (και όχι µόνο!) των 

γλωσσών προγραµµατισµού χρησιµοποιούµε γραµµατικές

• Οι γραµµατικές ορίζουν

– το ποια είναι τα επιτρεπτά προγράµµατα µιας γλώσσας, αλλά και

– το συντακτικό δένδρο για αυτά τα προγράµµατα– το συντακτικό δένδρο για αυτά τα προγράµµατα

– το δένδρο µε τη σειρά του ορίζει τη σειρά εκτέλεσης των εντολών

– και κατά συνέπεια συνεισφέρει στον ορισµό της σηµασιολογίας

• Υπάρχει ισχυρή σύνδεση µεταξύ θεωρίας και πράξης

– ∆ύο γραµµατικές, δύο µέρη του µεταγλωττιστή (compiler)

– Σηµείωση:Σηµείωση:Σηµείωση:Σηµείωση: Υπάρχουν προγράµµατα, γεννήτριες συντακτικών 
αναλυτών (parser generators), που µπορούν να δηµιουργήσουν 
αυτόµατα τον κώδικα του λεκτικού και του συντακτικού αναλυτή 
από τη γραµµατική µιας γλώσσας

